
Digital System Design

Lecture 4

Objectives

After completing this lecture, you will be able to:

 Describe the behavioral modeling structures

 Describe procedural constructs

 Understand the features of initial blocks

 Understand the features of always blocks

 Distinguish the differences between blocking and 

nonblocking assignments

 Understand the features of timing controls

 Understand the features of selection constructs

 Understand the features of loop constructs

Behavioral Modeling Structures

 Assignments: 

— continuous assignment (assign expression) (Dataflow modeling)

— blocking assignment  (=)

— nonblocking assignment (<=)

— assign … deassign

— force … release

 Selection structures: 

— if … else

— case (case, casex, casez)

 Iterative structures:

— repeat

— for

— while

— forever

Procedural Constructs

 Procedural Constructs 

— initial statements are used to initialize variables and set 
values into variables or nets.

— always statements are used to model the continuous 
operations required in the hardware modules.

— Each always statement corresponds to a piece of logic 
circuit.

— All other behavioral statements must be inside an initial or 
always block.

— initial and always statements:

 Each represents a separate activity flow.

 Each activity starts at simulation 0.

 They cannot be nested.

initial Statements

 An initial block

— is composed of all statements inside an initial statement. 

— executes exactly once during simulation.

— is used to initialize signals, or monitor waveforms, etc.

— starts to execute concurrently at simulation time 0 and 

finishes execution independently when multiple initial 

blocks exist.

reg  x, y, z;

initial begin // complex statement

x = 1`b0;  y = 1`b1;  z = 1`b0;

#10     x = 1`b1;  y = 1`b1;  z = 1`b1;

end

initial x = 1`b0;  // single statement

initial Statements

— Combined variable declaration and initialization

— Combined port/data declaration and initialization

reg  clk;              // regular declaration

initial clk = 0;

reg clk = 0;         // can be used only at module level

module adder(x, y, c , sum, c_out); 

input [3:0]           x, y;

input                    c_in;

output reg [3:0]   sum = 0;

output reg            c_out = 0;

module adder(input [3:0]           x, y,

input                    c_in,  

output reg [3:0]   sum = 0,

output reg            c_out = 0

);  // ANSI style

always Statements

 An always block

— consists of all behavioral statements inside an always 

statement. 

— starts at simulation time 0.

— executes continuously during simulation.

— is used to model a block of activity being repeated 

continuously in a digital circuit.

Q: What will be happened in the following statement?

reg  clock;                            // a clock generator

initial  clock = 1`b0;            // initial clock = 0

always  #5 clock = ~clock;  // period = 10

always  begin 

initial  clock = 1`b0; 

#5 clock =  ~clock;  

end

Procedural Assignments

 Procedural assignments 

— must be placed inside initial or always blocks. 

— update values of variable data types (reg, integer, real, or 
time.) 

variable_lvalue = [timing_control] expression

[timing_control] variable_lvalue = expression

— variable_lvalue can be:

 a reg

 integer,

 real,

 time, or 

 a memory element, 

 a bit select, a part select, a concatenation of any of the above.

Procedural Assignments

 The bit widths of both left-hand and right-hand sides 

need not be the same.

— The right-hand side is truncated if it has more bits.

 by keeping the least significant bits

— The right-hand side is filled with zeros in the most 

significant bits when it has fewer bits.

 Two types of procedural assignments:

— blocking: using the operator “=“

— nonblocking: using the operator “<=“

Blocking Assignments

 Blocking assignments

— are executed in the order they are specified.

— use the “=“ operator.

// an example illustrating blocking assignments

module blocking;

reg  x, y, z;

// blocking assignments

initial begin

x = #5 1'b0;    // x will be assigned 0 at time 5

y = #3 1'b1;    // y will be assigned 1 at time 8

z = #6 1'b0;    // z will be assigned 0 at time 14

end

endmodule

Blocking Assignments

module twos_adder_behavioral(x, y, c_in, sum, c_out); 

// I/O port declarations

input  [3:0] x, y;                // declare as a 4-bit array

input  c_in;

output reg  [3:0] sum;       // declare as a 4-bit array

output reg        c_out;

reg [3:0]        t;           // outputs of xor gates

// specify the function of a two's complement adder

always @(x, y, c_in) begin  // define two’s adder function

t = y ^ {4{c_in}};  // What is wrong with:  t = y ^ c_in ?

{c_out, sum} = x + t + c_in;

end

endmodule 

A register does not correspond to a 

memory element after synthesizing the 

circuit.

Q: What will happen if we change blocking operators (=) into nonblocking 

operators (<=)? Answer: t will be deferred one time unit.



Digital System Design

Lecture 4

Nonblocking Assignments

 Nonblocking assignments

— are executed without blocking the other statements.

— use the <= operator.

— are used to model several concurrent data transfers.

// an example illustrating nonblocking assignments

module nonblocking;

reg  x, y, z;

// nonblocking assignments

initial begin

x <= #5 1'b0;  // x will be assigned 0 at time 5

y <= #3 1'b1;  // y will be assigned 1 at time 3

z <= #6 1'b0;  // z will be assigned 0 at time 6

end

endmodule

Nonblocking Assignments

// an example of right-shift register without reset.

module shift_reg_4b(clk, din, qout);

input  clk;

input  din;

output reg [3:0] qout;

// the body of a 4-bit shift register

always @(posedge clk) 

qout <= {din, qout[3:1]};  // Right shift

endmodule

Q: What will happen if we change nonblocking operator (<=) into blocking 

operator (=)? 

Answer: Nothing will happen. Why ?

Try it in your system! Please

Race Conditions

// using blocking assignment statements

always @(posedge clock)   // has race condition

x = y;

always @(posedge clock)

y = x;

// using nonblocking assignment statements

always @(posedge clock)   // has no race condition

x <= y;

always @(posedge clock)

y <= x;

Note that: In simulation stage, three steps are performed for nonblocking statements:

1. Read the values of all right-hand-side variables;

2. Evaluate the right-hand-side expressions and store in temporary variables;

3. Assign the values stored in the temporary variables to the left-hand-side variables.

Blocking vs. Nonblocking 
Assignments

// shift register module example --- an incorrect implementation

module shift_reg_blocking(clk, sin, qout);

input  clk; 

input  sin; // serial data input 

output reg [3:0] qout;

// The body of a 4-bit shift register

always @(posedge clk)

begin      // using blocking assignments

qout[0] = sin;

qout[1] = qout[0];

qout[2] = qout[1];

qout[3] = qout[2];

end

endmodule

Note that: When using qout = {qout[2:0], sin} instead of qout <= {qout[2:0], sin}, the 

result will not be different. Why? Explain it.

qout[3:0]

qout[3:0]
[3:0]

clk

sin

[3:0]
Q[3:0]D[3:0]

Blocking vs. Nonblocking 
Assignments

// shift register module example ---a correct implementation

module shift_reg_nonblocking(clk, sin, qout);

input  clk; 

input  sin; // serial data input

output reg [3:0] qout;

// The body of a 4-bit shift register

always @(posedge clk)

begin                             // using nonblocking assignments

qout[0] <= sin;           // It is even better to use

qout[1] <= qout[0];    // qout <= {qout[2:0], sin};

qout[2] <= qout[1];

qout[3] <= qout[2];

end

endmodule

qout[3:0]

qout[3:0]
[3:0]

clk

sin

[2:0] [3:0]
Q[3:0]D[3:0]

Blocking vs. Nonblocking 
Assignments

 Consider the difference between the following two 

always blocks. Assume that the value of count is 1

and finish is 0 before entering the always block.

always @(posedge clk) begin: block_a

count = count – 1;

if (count == 0) finish = 1;

end

always @(posedge clk) begin: block_b

count <= count – 1;

if (count == 0) finish <= 1;

end

Result: finish = 1. 

(Different from that of gate-level.)
Result: finish = 0. 

(Same as that of gate-level.)

Coding Style for Blocking / 
Nonblocking Assignments

 Coding style: In the always block

— Use nonblocking operators (<=) when it is a piece of 

sequential logic; 

Otherwise, the result of RTL behavioral may be inconsistent with 

that of gate-level.

— Use blocking operators (=) when it is a piece of 

combinational logic.

4-18



Digital System Design

Lecture 4

Timing Controls

 Timing controls specify the simulation time at which 
procedural statements will be executed.

 In Verilog HDL, if there are no timing control 
statements, the simulation time will not advance.

 Timing Controls

— Delay timing control

Regular delay control

 Intra assignment delay control

— Event timing control

 Edge-triggered event control

Named event control 

Event or control

 Level-sensitive event control

Regular Delay Control

 Regular delay control

— A non-zero delay is specified to the left of a procedural 

assignment.

— It defers the execution of the entire statement.

4-20

reg  x, y;

integer count;

// The “<=” operators in the following statements can be replaced with “=”

// without affecting the results.

#25         y <= ~x;                  // execute at time 25

#15  count <= count  + 1;      // execute at time 40

Intra-Assignment Delay Control

 Intra-assignment delay control

— A non-zero delay is specified to the right of the 

assignment operator.

— It defers the assignment to the left-hand-side variable.

y = #25  ~x;              // evaluate at time 0 but assign to y at time 25

count = #15 count  + 1;   // evaluate at time 0 but assign to count at time 40

y <= #25  ~x;              // evaluate at time 0 but assign to y at time 25

count <= #15 count  + 1;   // evaluate at time 0 but assign to count at time 15

Event Timing Control

 Event Timing Control

— An event is the change in the value on a variable or a net. 

— The execution of a procedural statement can be 

synchronized with an event. 

 Two types of event control

— Edge-triggered event control

Named event control

 Event or control

— Level-sensitive event control

Edge-Triggered Event Control

 Edge-triggered event control

— The symbol @ is used to specify such event control.

@(posedge clock): at the positive edge 

@(negedge clock): at the negative edge

always @(posedge clock)  begin

reg1 <= #25  in_1;                                   // intra-assignment delay control

reg2 <= @(negedge clock) in_2 ^ in_3;  // edge-triggered event control

reg3 <= in_1;                                           // no delay control

end

Named Event Control

 A named event

— is declared with the keyword event.

— does not hold any data.

— is triggered by the symbol ->. 

— is recognized by the symbol @.

event  received_data;  // declare an event received_data

// trigger event received_data

always @(posedge clock)  if (last_byte) -> received_data;  

always @(received_data) begin ….. end // execute event-dependent operations



Digital System Design

Lecture 4

Event or Control

 Event or control

— uses the keyword or to specify multiple triggers.

— can be replaced by the “,”.

— can use @* or @(*) to mean a change on any signal.

always @(posedge clock or negative reset_n)  // event or control

begin

if (!reset_n) q <= 1`b0;    // asynchronous reset.

else              q <= d;

end

Level-Sensitive Event Control

 Level-sensitive event control

— uses the keyword wait.

always 

wait (count_enable) count = count –1 ;

always 

wait (count_enable) #10 count = count –1 ;

Selection Constructs

 Selection structures

— make a selection according to the given condition.

 Two types

— if…else statement

— case statement

 if…else statement syntax

— if statement only 

— if and one else statement

— nested if-else-if statement

 The else part is always associated to the closest 

previous if that lacks an else.

Selection Constructs

if (<expression>)  true_statement ;

if (<expression>) true_statement; else false_statement;

if (<expression1>) true_statement1; 

else if (<expression2>) true_statement2; 

else false_statement;

Selection Constructs

module mux4_to_1_ifelse (i0, i1, i2, i3, s1, s0, out);

// port declarations

input    i0, i1, i2, i3;

input    s1, s0;

output  reg out;

// using conditional operator if  else statement

always @(*)  // triggered for all signals used in the if else statement

if (s1) begin

if (s0)  out = i3; else out = i2; end

else begin

if (s0)  out = i1; else out = i0; end

endmodule

 un1_s1_1 

 un1_s1_2 

 un1_s0_1 

 un1_s0_2 

 out 

e

d

e

d

e

d

e

d

out

s0

s1

i3

i2

i1

i0

A Simple 4-bit Counter

module counter (clock, clear, qout);

input   clock, clear;

output reg [3:0] qout;

// the body of the 4-bit counter.

always @(negedge clock or posedge clear)

begin

if (clear)

qout <= 4'd0;

else      

qout <= (qout + 1) ;   // qout = (qout + 1) % 16;

end

endmodule

 un3_qout[3:0] 

+

 qout[3:0] 

R

[3:0]

[3:0]

1

[3:0]
Q[3:0]

[3:0]
D[3:0]

clear

clock

qout[3:0]
[3:0]

Selection Constructs

 case statement: a multiway selection.

— compares the expression to the alternatives in the order 
they are written. 

— compares 0, 1, x, and z values in the expression and the 
alternative bit for bit. 

— executes the default statement if no matches are made.

— fills zeros to match the unequal bit widths between the 
expression and the alternative.

— is acted like a multiplexer.

 The default statement is optional and at most one 
default statement can be placed inside one case 
statement.

 A block statement must be grouped by begin and 
end.

A 4-to-1 MUX Example

// a 4-to-1 multiplexer using case statement

module mux_4x1_case (I0, I1, I2, I3, S, Y); 

input I0, I1, I2, I3;

input [1:0] S;    // declare S as a two-bit selection signal.

output reg  Y;

always @(I0 or I1 or I2 or I3 or S) // It can use always @(*).

case (S)  

2'b00: Y = I0;  

2'b01: Y = I1;

2'b10: Y = I2;  

2'b11: Y = I3;

endcase

endmodule

Y9

 un1_S_2 

 un1_S_3 

 un1_S_4 

 Y 

e

d

e

d

e

d

e

d

[0]

[1]

[0]

[1]

[0]

[1]

[1]

[0]

Y

S[1:0]
[1:0]

I3

I2

I1

I0

A 4-to-1 MUX Example

// a 4-to-1 multiplexer using case and default statements.

module mux4_to_1_case_default (i0, i1, i2, i3, s1, s0, out); 

input   i0, i1, i2, i3, s1, s0;

output reg out;       //output declared as register

always @(s1 or s0 or i0 or i1 or i2 or i3)

case ({s1, s0})   // concatenate s1 and s0 as a two-bit selection signals

2'b00: out = i0;

2'b01: out = i1;

2'b10: out = i2;

2'b11: out = i3;

default: out = 1'bx;  // using default to include all other possible cases.

endcase

endmodule

Selection Constructs

 casex and casez statements

— are used to perform a multiway selection like that of case 

statement. 

— compare only non-x or z positions in the case expression 

and the case alternatives.

 casez treats all z values as don’t cares. 

 casex treats all x and z values as don’t cares.

Selection Constructs

// an example illustrating how to count the trailing zeros in a nibble.

module trailing_zero_4b (data, out); 

input   [3:0] data;

output reg [2:0] out;   //output declared as register

always @(data)

casex (data)            // treat both x and z as don’t care conditions.

4'bxxx1: out = 0;

4'bxx10: out = 1;

4'bx100: out = 2;

4'b1000: out = 3;

4'b0000: out = 4;

default:  out = 3'b111; //using default to include all other possible cases.

endcase

endmodule

Loop Constructs

 Loop constructs control the execution of a statement 

zero, one, or more times. 

 Loop constructs

— can appear only inside an initial or always block. 

— may contain delay expressions.

 Four types

— while loop executes a statement until an expression 

becomes false. 

— for loop repeatedly executes a statement.

— repeat loop executes a statement a fixed number of times.

— forever loop continuously executes a statement.

The while Loop Structure

 A while loop

— executes until the condition is false.

— shall not be executed at all if the condition_expr starts out 

false.

while (condition_expr) statement;

while (count < 12) count <= count + 1;

while (count <= 100 && flag) begin

// put statements wanted to be carried out here.

end

The while Loop Structure

// an example illustrating how to count the zeros in a byte.

module zero_count_while (data, out); 

input   [7:0] data;

output reg [3:0] out;        //output declared as register

integer i;

always @(data) begin

out = 0; i = 0;

while (i <= 7) begin     // simple condition

if (data[i] == 0) out = out + 1; // may be replaced with out = out + ~data[i].

i = i + 1; end

end

endmodule

Note that: If we replace the blocking assignments by nonblocking assignments, 

the resulting program will not work properly. Why ? Try it! Explain it.

The while Loop Structure

// an example illustrating how to count the trailing zeros in a byte.

module trailing_zero_while (data, out); 

input   [7:0] data;

output reg [3:0] out;  // output declared as register

integer i;                    // loop counter

always @(data) begin

out = 0; i = 0;

while (data[i] == 0 && i <= 7) begin // complex condition

out = out + 1;

i = i + 1;

end

end

endmodule

Note that: Please distinguish the difference between this example and the 

previous one.

The for Loop Structure

 A for loop is used to perform a counting loop. It

— behaves like the for statement in C programming language.

— is equivalent to

for (init_expr; condition_expr; update_expr) statement;

init_expr;

while (condition_expr) begin

statement;

update_expr;

end

The for Loop Structure

// an example illustrating how to count the zeros in a byte.

module zero_count_for (data, out); 

input   [7:0] data;

output reg [3:0] out;   // output declared as register

integer i;

always @(data) begin

out = 0;

for (i = 0; i <= 7; i = i + 1) // simple condition

if (data[i] == 0)     

out = out + 1;           // may be replaced with out = out + ~data[i]. 

end

endmodule

Note that: If we replace the blocking assignments by nonblocking assignments, 

the resulting program will not work properly. Why ? Try it! Explain it.

The for Loop Structure

// an example illustrating how to count the trailing zeros in a byte.

module trailing_zero_for (data, out); 

input   [7:0] data;

output reg [3:0] out;  // output declared as register

integer i;                    // loop counter

always @(data) begin

out = 0;

for (i = 0; data[i] == 0 && i <= 7; i = i + 1) // complex condition

out = out + 1;

end

endmodule

Note that: Please distinguish the difference between this example and the 

previous one.

The repeat Loop Structure

 A repeat loop performs a loop a fixed number of 

times.

— counter_expr can be a constant, a variable or a signal 

value.

— counter_expr is evaluated only once before starting the 

execution of statement (loop).

repeat (counter_expr) statement;

The repeat Loop Structure

— Examples:

i = 0;

repeat (32) begin 

state[i] = 0;      // initialize to zeros

i  =  i  +  1;       // next item

end

repeat (cycles) begin                           // cycles must be evaluated to a number

@(posedge clock) buffer[i] <= data;  // before entering the loop.

i  <=  i  + 1;      // next item

end

The forever Loop Structure

 A forever loop continuously performs a loop until 

the $finish task is encountered. It

— is equivalent to a while loop with an always true 

expression such as while (1).

— can be exited by the use of disable statement.

forever statement;



Digital System Design

Lecture 4

The forever Loop Structure

— The forever statement example

— The forever statement is usually used with timing control 

statements.

initial  begin

clock <= 0;

forever begin

#10 clock <= 1; 

#5   clock <= 0;

end

end

reg clock, x, y;

initial

forever @(posedge clock) x <= y;


